

Rules for integrands of the form $(a + b x)^m (c + d x)^n$

1. $\int (a + b x)^m (c + d x)^n dx$ when $b c - a d \neq 0 \wedge m + n + 2 = 0$

1. $\int \frac{1}{(a + b x) (c + d x)} dx$ when $b c - a d \neq 0$

1: $\int \frac{1}{(a + b x) (c + d x)} dx$ when $b c + a d = 0$

Derivation: Algebraic simplification

Basis: If $b c + a d = 0$, then $(a + b x) (c + d x) = a c + b d x^2$

Rule 1.1.1.2.1.1.1: If $b c + a d = 0$, then

$$\int \frac{1}{(a + b x) (c + d x)} dx \rightarrow \int \frac{1}{a c + b d x^2} dx$$

Program code:

```
Int[1/((a+b_.x_)*(c+d_.x_)),x_Symbol] :=  
  Int[1/(a*c+b*d*x^2),x] /;  
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0]
```

$$2: \int \frac{1}{(a+bx)(c+dx)} dx \text{ when } bc - ad \neq 0$$

Derivation: Algebraic expansion

$$\text{Basis: } \frac{1}{(a+bx)(c+dx)} = \frac{b}{(bc-ad)(a+bx)} - \frac{d}{(bc-ad)(c+dx)}$$

Rule 1.1.1.2.1.1.2: If $bc - ad \neq 0$, then

$$\int \frac{1}{(a+bx)(c+dx)} dx \rightarrow \frac{b}{bc-ad} \int \frac{1}{a+bx} dx - \frac{d}{bc-ad} \int \frac{1}{c+dx} dx$$

Program code:

```
Int[1/((a_+b_*x_)*(c_+d_*x_)),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*x),x] - d/(b*c-a*d)*Int[1/(c+d*x),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2: $\int (a+bx)^m (c+dx)^n dx$ when $bc - ad \neq 0 \wedge m+n+2 \neq 0 \wedge m \neq -1$

Reference: G&R 2.155, CRC 59a with $m+n+2 \neq 0$

Reference: G&R 2.110.2 or 2.110.6 with $k=1$ and $m+n+2 \neq 0$

Derivation: Linear recurrence 3 with $m+n+2 \neq 0$

Rule 1.1.1.2.1.2: If $bc - ad \neq 0 \wedge m+n+2 \neq 0 \wedge m \neq -1$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1} (c+dx)^{n+1}}{(bc-ad)(m+1)}$$

Program code:

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/((b*c-a*d)*(m+1))/;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && EqQ[m+n+2,0] && NeQ[m,-1]
```

$$2. \int (a+bx)^m (c+dx)^n dx \text{ when } bc+ad=0$$

$$1. \int (a+bx)^m (c+dx)^n dx \text{ when } bc+ad=0 \wedge m-n=0$$

$$1: \int (a+bx)^m (c+dx)^m dx \text{ when } bc+ad=0 \wedge m+\frac{1}{2} \in \mathbb{Z}^+$$

Derivation: Inverted integration by parts

Rule 1.1.1.2.2.1.1: If $bc+ad=0 \wedge m+\frac{1}{2} \in \mathbb{Z}^+$, then

$$\int (a+bx)^m (c+dx)^m dx \rightarrow \frac{x(a+bx)^m (c+dx)^m}{2m+1} + \frac{2acm}{2m+1} \int (a+bx)^{m-1} (c+dx)^{m-1} dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  x*(a+b*x)^m*(c+d*x)^m/(2*m+1) + 2*a*c*m/(2*m+1)*Int[(a+b*x)^(m-1)*(c+d*x)^(m-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && IGtQ[m+1/2,0]
```

$$2. \int (a+bx)^m (c+dx)^m dx \text{ when } bc+ad=0 \wedge m+\frac{1}{2} \in \mathbb{Z}^-$$

$$1: \int \frac{1}{(a+bx)^{3/2} (c+dx)^{3/2}} dx \text{ when } bc+ad=0$$

Rule 1.1.1.2.2.1.2.1: If $bc+ad=0$, then

$$\int \frac{1}{(a+bx)^{3/2} (c+dx)^{3/2}} dx \rightarrow \frac{x}{ac\sqrt{a+bx}\sqrt{c+dx}}$$

Program code:

```
Int[1/((a_+b_.*x_)^(3/2)*(c_+d_.*x_)^(3/2)),x_Symbol] :=
  x/(a*c*Sqrt[a+b*x]*Sqrt[c+d*x]) /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0]
```

$$2: \int (a+bx)^m (c+dx)^m dx \text{ when } bc+ad=0 \wedge m+\frac{3}{2} \in \mathbb{Z}^-$$

Derivation: Integration by parts

$$\text{Basis: } (a+bx)^m (c+dx)^m = x^{2(m+1)+1} \frac{(a+bx)^m (c+dx)^m}{x^{2(m+1)+1}}$$

$$\text{Basis: If } bc+ad=0, \text{ then } \int \frac{(a+bx)^m (c+dx)^m}{x^{2(m+1)+1}} dx = -\frac{(a+bx)^{m+1} (c+dx)^{m+1}}{x^{2(m+1)} 2ac(m+1)}$$

Rule 1.1.1.2.2.1.2.2: If $bc+ad=0 \wedge m+\frac{3}{2} \in \mathbb{Z}^-$, then

$$\int (a+bx)^m (c+dx)^m dx \rightarrow -\frac{x(a+bx)^{m+1} (c+dx)^{m+1}}{2ac(m+1)} + \frac{2m+3}{2ac(m+1)} \int (a+bx)^{m+1} (c+dx)^{m+1} dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
-x*(a+b*x)^(m+1)*(c+d*x)^(m+1)/(2*a*c*(m+1)) +
(2*m+3)/(2*a*c*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(m+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && ILtQ[m+3/2,0]
```

$$3: \int (a+bx)^m (c+dx)^m dx \text{ when } bc+ad=0 \wedge (m \in \mathbb{Z} \vee a>0 \wedge c>0)$$

Derivation: Algebraic simplification

Basis: If $bc+ad=0 \wedge (m \in \mathbb{Z} \vee a>0 \wedge c>0)$, then $(a+bx)^m (c+dx)^m = (ac+bdx^2)^m$

Rule 1.1.1.2.2.1.3: If $bc+ad=0 \wedge (m \in \mathbb{Z} \vee a>0 \wedge c>0)$, then

$$\int (a+bx)^m (c+dx)^m dx \rightarrow \int (ac+bdx^2)^m dx$$

Program code:

```
Int[(a+b_*x_)^m_.*(c+d_*x_)^m_.,x_Symbol] :=
  Int[(a*c+b*d*x^2)^m,x] /;
  FreeQ[{a,b,c,d,m},x] && EqQ[b*c+a*d,0] && (IntegerQ[m] || GtQ[a,0] && GtQ[c,0])
```

$$4. \int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \text{ when } bc+ad=0$$

$$1: \int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \text{ when } bc+ad=0 \wedge a>0 \wedge a+c=0$$

Rule 1.1.1.2.2.1.4.1: If $bc+ad=0 \wedge a>0 \wedge a+c=0$, then

$$\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \rightarrow \frac{1}{b} \text{ArcCosh}\left[\frac{bx}{a}\right]$$

Program code:

```
Int[1/(Sqrt[a+b_*x_]*Sqrt[c+d_*x_]),x_Symbol] :=
  ArcCosh[b*x/a]/b /;
  FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && GtQ[a,0] && EqQ[a+c,0]
```

$$2: \int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \text{ when } bc + ad = 0$$

Derivation: Integration by substitution

$$\text{Basis: } \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} = 2 \text{ Subst} \left[\frac{1}{b-dx^2}, x, \frac{\sqrt{a+bx}}{\sqrt{c+dx}} \right] \partial_x \frac{\sqrt{a+bx}}{\sqrt{c+dx}}$$

Rule 1.1.1.2.2.1.4.2: If $bc + ad = 0$, then

$$\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \rightarrow 2 \text{ Subst} \left[\int \frac{1}{b-dx^2} dx, x, \frac{\sqrt{a+bx}}{\sqrt{c+dx}} \right]$$

Program code:

```
Int[1/(Sqrt[a+_.*x_]*Sqrt[c+_.*x_]),x_Symbol] :=
  2*Subst[Int[1/(b-d*x^2),x],x,Sqrt[a+b*x]/Sqrt[c+d*x]] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0]
```

5: $\int (a+bx)^m (c+dx)^m dx$ when $bc+ad = 0 \wedge 2m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $bc+ad = 0$, then $\partial_x \frac{(a+bx)^m (c+dx)^m}{(ac+bdx^2)^m} = 0$

Basis: If $bc+ad = 0$, then $\frac{(a+bx)^m (c+dx)^m}{(ac+bdx^2)^m} = \frac{(a+bx)^{\text{FracPart}[m]} (c+dx)^{\text{FracPart}[m]}}{(ac+bdx^2)^{\text{FracPart}[m]}}$

Rule 1.1.1.2.2.1.5: If $bc+ad = 0 \wedge 2m \notin \mathbb{Z}$, then

$$\int (a+bx)^m (c+dx)^m dx \rightarrow \frac{(a+bx)^{\text{FracPart}[m]} (c+dx)^{\text{FracPart}[m]}}{(ac+bdx^2)^{\text{FracPart}[m]}} \int (ac+bdx^2)^m dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  (a+b*x)^FracPart[m]*(c+d*x)^FracPart[m]/(a*c+b*d*x^2)^FracPart[m]*Int[(a*c+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b*c+a*d,0] && Not[IntegerQ[2*m]]
```


$$2: \int \frac{1}{(a+bx)^{5/4} (c+dx)^{1/4}} dx \text{ when } bc+ad=0 \wedge a^2 b^2 < 0$$

Derivation: Linear recurrence 2

Derivation: Inverted integration by parts

Rule 1.1.1.2.2.2: If $bc+ad=0 \wedge a^2 b^2 < 0$, then

$$\int \frac{1}{(a+bx)^{5/4} (c+dx)^{1/4}} dx \rightarrow -\frac{2}{b(a+bx)^{1/4} (c+dx)^{1/4}} + c \int \frac{1}{(a+bx)^{5/4} (c+dx)^{5/4}} dx$$

Program code:

```
Int[1/((a+b.*x_)^(5/4)*(c+d.*x_)^(1/4)),x_Symbol] :=
-2/(b*(a+b*x)^(1/4)*(c+d*x)^(1/4)) + c*Int[1/((a+b*x)^(5/4)*(c+d*x)^(5/4)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && NegQ[a^2*b^2]
```

$$3: \int \frac{1}{(a+bx)^{9/4} (c+dx)^{1/4}} dx \text{ when } bc+ad=0 \wedge a^2 b^2 < 0$$

Derivation: Linear recurrence 1

Rule 1.1.1.2.2.3: If $bc+ad=0 \wedge a^2 b^2 < 0$, then

$$\int \frac{1}{(a+bx)^{9/4} (c+dx)^{1/4}} dx \rightarrow -\frac{4}{5b(a+bx)^{5/4} (c+dx)^{1/4}} - \frac{d}{5b} \int \frac{1}{(a+bx)^{5/4} (c+dx)^{5/4}} dx$$

Program code:

```
Int[1/((a+b.*x_)^(9/4)*(c+d.*x_)^(1/4)),x_Symbol] :=
-4/(5*b*(a+b*x)^(5/4)*(c+d*x)^(1/4)) - d/(5*b)*Int[1/((a+b*x)^(5/4)*(c+d*x)^(5/4)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && NegQ[a^2*b^2]
```

4: $\int (a+bx)^m (c+dx)^n dx$ when $bc+ad=0 \wedge m+\frac{1}{2} \in \mathbb{Z}^+ \wedge n+\frac{1}{2} \in \mathbb{Z}^+ \wedge m < n$

Rule 1.1.1.2.2.4: If $bc+ad=0 \wedge m+\frac{1}{2} \in \mathbb{Z}^+ \wedge n+\frac{1}{2} \in \mathbb{Z}^+ \wedge m < n$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1} (c+dx)^n}{b(m+n+1)} + \frac{2cn}{m+n+1} \int (a+bx)^m (c+dx)^{n-1} dx$$

Program code:

```
Int[(a+b*x)^m*(c+d*x)^n,x_Symbol] :=
(a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) +
2*c*n/(m+n+1)*Int[(a+b*x)^m*(c+d*x)^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && IGtQ[m+1/2,0] && IGtQ[n+1/2,0] && LtQ[m,n]
```

5: $\int (a+bx)^m (c+dx)^n dx$ when $bc+ad=0 \wedge m-\frac{1}{2} \in \mathbb{Z}^- \wedge n-\frac{1}{2} \in \mathbb{Z}^- \wedge m < n$

Rule 1.1.1.2.2.5: If $bc+ad=0 \wedge m-\frac{1}{2} \in \mathbb{Z}^- \wedge n-\frac{1}{2} \in \mathbb{Z}^- \wedge m < n$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow -\frac{(a+bx)^{m+1} (c+dx)^{n+1}}{2ad(m+1)} + \frac{m+n+2}{2a(m+1)} \int (a+bx)^{m+1} (c+dx)^n dx$$

Program code:

```
Int[(a+b*x)^m*(c+d*x)^n,x_Symbol] :=
-(a+b*x)^(m+1)*(c+d*x)^(n+1)/(2*a*d*(m+1)) +
(m+n+2)/(2*a*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && ILtQ[m-1/2,0] && ILtQ[n-1/2,0] && LtQ[m,n]
```

3: $\int (a+bx)^m (c+dx)^n dx$ when $bc - ad \neq 0 \wedge m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.1.1.2.3: If $bc - ad \neq 0 \wedge m \in \mathbb{Z}$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \int \text{ExpandIntegrand}[(a+bx)^m (c+dx)^n, x] dx$$

Program code:

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && IGtQ[m,0] &&
  (Not[IntegerQ[n]] || EqQ[c,0] && LeQ[7*m+4*n+4,0] || LtQ[9*m+5*(n+1),0] || GtQ[m+n+2,0])
```

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && ILtQ[m,0] && IntegerQ[n] && Not[IGtQ[n,0] && LtQ[m+n+2,0]]
```

$$4. \int \frac{(c+dx)^n}{a+bx} dx \text{ when } bc - ad \neq 0$$

$$1: \int \frac{(c+dx)^n}{a+bx} dx \text{ when } bc - ad \neq 0 \wedge n > 0$$

Derivation: Integration by parts

Derivation: Algebraic expansion

$$\text{Basis: } \frac{(c+dx)^n}{a+bx} = \frac{d(c+dx)^{n-1}}{b} + \frac{(bc-ad)(c+dx)^{n-1}}{b(a+bx)}$$

Rule 1.1.1.2.4.1: If $bc - ad \neq 0 \wedge n > 0$, then

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow \frac{(c+dx)^n}{bn} + \frac{bc-ad}{b} \int \frac{(c+dx)^{n-1}}{a+bx} dx$$

Program code:

```
Int[(c_.+d_.*x_)^n/(a_.+b_.*x_),x_Symbol] :=
  (c+d*x)^n/(b*n) +
  (b*c-a*d)/b*Int[(c+d*x)^(n-1)/(a+b*x),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && GtQ[n,0]
```

$$2: \int \frac{(c+dx)^n}{a+bx} dx \text{ when } bc - ad \neq 0 \wedge n < -1$$

Derivation: Integration by parts

Derivation: Algebraic expansion

$$\text{Basis: } \frac{(c+dx)^n}{a+bx} = -\frac{d(c+dx)^n}{bc-ad} + \frac{b(n+1)(c+dx)^{n+1}}{(n+1)(bc-ad)(a+bx)}$$

Rule 1.1.1.2.4.2: If $bc - ad \neq 0 \wedge n < -1$, then

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow -\frac{(c+dx)^{n+1}}{(n+1)(bc-ad)} + \frac{b(n+1)}{(n+1)(bc-ad)} \int \frac{(c+dx)^{n+1}}{a+bx} dx$$

Program code:

```
Int[(c_.+d_.*x_)^n/(a_.+b_.*x_),x_Symbol] :=
  -(c+d*x)^(n+1)/((n+1)*(b*c-a*d)) +
  b*(n+1)/((n+1)*(b*c-a*d))*Int[(c+d*x)^(n+1)/(a+b*x),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[n,-1]
```

$$3. \int \frac{(c+dx)^n}{a+bx} dx \text{ when } bc-ad \neq 0 \wedge -1 < n < 0$$

$$1. \int \frac{1}{(a+bx)(c+dx)^{1/3}} dx \text{ when } bc-ad \neq 0$$

$$1: \int \frac{1}{(a+bx)(c+dx)^{1/3}} dx \text{ when } \frac{bc-ad}{b} > 0$$

Derivation: Integration by substitution

Basis: Let $q = \left(\frac{bc-ad}{b}\right)^{1/3}$, then $\frac{1}{(a+bx)(c+dx)^{1/3}} = -\frac{1}{2q(a+bx)} - \text{Subst}\left[\frac{3}{2bq(q-x)} - \frac{3}{2b(q^2+qx+x^2)}, x, (c+dx)^{1/3}\right] \partial_x (c+dx)^{1/3}$

Rule 1.1.1.2.4.3.1.1: If $\frac{bc-ad}{b} > 0$, let $q = \left(\frac{bc-ad}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+bx)(c+dx)^{1/3}} dx \rightarrow$$

$$-\frac{\text{Log}[a+bx]}{2bq} - \frac{3}{2bq} \text{Subst}\left[\int \frac{1}{q-x} dx, x, (c+dx)^{1/3}\right] + \frac{3}{2b} \text{Subst}\left[\int \frac{1}{q^2+qx+x^2} dx, x, (c+dx)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(1/3)),x_Symbol] :=
  With[{q=Rt[(b*c-a*d)/b,3]},
  -Log[RemoveContent[a+b*x,x]]/(2*b*q) -
  3/(2*b*q)*Subst[Int[1/(q-x),x],x,(c+d*x)^(1/3)] +
  3/(2*b)*Subst[Int[1/(q^2+q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
FreeQ[{a,b,c,d},x] && PosQ[(b*c-a*d)/b]
```

$$2: \int \frac{1}{(a+bx)(c+dx)^{1/3}} dx \text{ when } \frac{bc-ad}{b} \neq 0$$

Derivation: Integration by substitution

Basis: Let $q = \left(-\frac{bc-ad}{b}\right)^{1/3}$, then $\frac{1}{(a+bx)(c+dx)^{1/3}} = \frac{1}{2q(a+bx)} - \text{Subst}\left[\frac{3}{2bq(q+x)} - \frac{3}{2b(q^2-qx+x^2)}, x, (c+dx)^{1/3}\right] \partial_x (c+dx)^{1/3}$

Rule 1.1.1.2.4.3.1.2: If $\frac{bc-ad}{b} \neq 0$, let $q = \left(-\frac{bc-ad}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+bx)(c+dx)^{1/3}} dx \rightarrow \frac{\text{Log}[a+bx]}{2bq} - \frac{3}{2bq} \text{Subst}\left[\int \frac{1}{q+x} dx, x, (c+dx)^{1/3}\right] + \frac{3}{2b} \text{Subst}\left[\int \frac{1}{q^2-qx+x^2} dx, x, (c+dx)^{1/3}\right]$$

Program code:

```
Int[1/((a_+b_*x_)*(c_+d_*x_)^(1/3)),x_Symbol] :=
  With[{q=Rt[-(b*c-a*d)/b,3]},
    Log[RemoveContent[a+b*x,x]]/(2*b*q) -
    3/(2*b*q)*Subst[Int[1/(q+x),x],x,(c+d*x)^(1/3)] +
    3/(2*b)*Subst[Int[1/(q^2-q*x+x^2),x],x,(c+d*x)^(1/3)] /;
    FreeQ[{a,b,c,d},x] && NegQ[(b*c-a*d)/b]
```

$$2. \int \frac{1}{(a+bx)(c+dx)^{2/3}} dx \text{ when } bc-ad \neq 0$$

$$1: \int \frac{1}{(a+bx)(c+dx)^{2/3}} dx \text{ when } \frac{bc-ad}{b} > 0$$

Derivation: Integration by substitution

Basis: Let $q = \left(\frac{bc-ad}{b}\right)^{1/3}$, then $\frac{1}{(a+bx)(c+dx)^{2/3}} = -\frac{1}{2q^2(a+bx)} - \text{Subst}\left[\frac{3}{2bq^2(q-x)} + \frac{3}{2bq(q^2+qx+x^2)}, x, (c+dx)^{1/3}\right] \partial_x (c+dx)^{1/3}$

Rule 1.1.1.2.4.3.2.1: If $\frac{bc-ad}{b} > 0$, let $q = \left(\frac{bc-ad}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+bx)(c+dx)^{2/3}} dx \rightarrow$$

$$-\frac{\text{Log}[a+bx]}{2bq^2} - \frac{3}{2bq^2} \text{Subst}\left[\int \frac{1}{q-x} dx, x, (c+dx)^{1/3}\right] - \frac{3}{2bq} \text{Subst}\left[\int \frac{1}{q^2+qx+x^2} dx, x, (c+dx)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[(b*c-a*d)/b,3]},
    -Log[RemoveContent[a+b*x,x]]/(2*b*q^2) -
    3/(2*b*q^2)*Subst[Int[1/(q-x),x],x,(c+d*x)^(1/3)] -
    3/(2*b*q)*Subst[Int[1/(q^2+q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
  FreeQ[{a,b,c,d},x] && PosQ[(b*c-a*d)/b]
```

2: $\int \frac{1}{(a+bx)(c+dx)^{2/3}} dx$ when $\frac{b*c-a*d}{b} \neq 0$

Derivation: Integration by substitution

Basis: Let $q = \left(-\frac{b*c-a*d}{b}\right)^{1/3}$, then $\frac{1}{(a+bx)(c+dx)^{2/3}} = -\frac{1}{2q^2(a+bx)} + \text{Subst}\left[\frac{3}{2bq^2(q+x)} + \frac{3}{2bq(q^2-qx+x^2)}, x, (c+dx)^{1/3}\right] \partial_x (c+dx)^{1/3}$

Rule 1.1.1.2.4.3.2.2: If $\frac{b*c-a*d}{b} \neq 0$, let $q = \left(-\frac{b*c-a*d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+bx)(c+dx)^{2/3}} dx \rightarrow$$

$$-\frac{\text{Log}[a+bx]}{2bq^2} + \frac{3}{2bq^2} \text{Subst}\left[\int \frac{1}{q+x} dx, x, (c+dx)^{1/3}\right] + \frac{3}{2bq} \text{Subst}\left[\int \frac{1}{q^2-qx+x^2} dx, x, (c+dx)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[-(b*c-a*d)/b,3]},
    -Log[RemoveContent[a+b*x,x]]/(2*b*q^2) +
    3/(2*b*q^2)*Subst[Int[1/(q+x),x],x,(c+d*x)^(1/3)] +
    3/(2*b*q)*Subst[Int[1/(q^2-q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
  FreeQ[{a,b,c,d},x] && NegQ[(b*c-a*d)/b]
```

$$3: \int \frac{(c+dx)^n}{a+bx} dx \text{ when } bc - ad \neq 0 \wedge -1 < n < 0$$

Derivation: Integration by substitution

Basis: If $p \in \mathbb{Z}^+$, then $\frac{(c+dx)^n}{a+bx} = p \text{ Subst} \left[\frac{x^{p(n+1)-1}}{ad-bc+bx^p}, x, (c+dx)^{1/p} \right] \partial_x (c+dx)^{1/p}$

Rule 1.1.1.2.4.3.3: If $bc - ad \neq 0 \wedge -1 < n < 0$, let $p = \text{Denominator}[n]$, then

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow p \text{ Subst} \left[\int \frac{x^{p(n+1)-1}}{ad-bc+bx^p} dx, x, (c+dx)^{1/p} \right]$$

Program code:

```
Int[(c_+d_*x_)^n/(a_+b_*x_),x_Symbol] :=
  With[{p=Denominator[n]},
    p*Subst[Int[x^(p*(n+1)-1)/(a*d-b*c+b*x^p),x],x,(c+d*x)^(1/p)] /;
    FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && GtQ[n,-1] && LtQ[n,0]
```


H: $\int \frac{(c+dx)^n}{a+bx} dx$ when $bc - ad \neq 0 \wedge n \notin \mathbb{Z}$

Rule 1.1.1.2.H.4: If $bc - ad \neq 0 \wedge n \notin \mathbb{Z}$, then

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow -\frac{(c+dx)^{n+1}}{(n+1)(bc-ad)} \text{Hypergeometric2F1}\left[1, n+1, n+2, \frac{b(c+dx)}{bc-ad}\right]$$

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow \frac{(c+dx)^{n+1}}{dn(a+bx)} \text{Hypergeometric2F1}\left[1, 1, 1-n, -\frac{bc-ad}{d(a+bx)}\right]$$

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow \frac{(c+dx)^n}{bn} \text{Hypergeometric2F1}\left[1, -n, 1-n, \frac{bc-ad}{b(c+dx)}\right]$$

$$\int \frac{(c+dx)^n}{a+bx} dx \rightarrow \frac{(c+dx)^n}{bn \left(\frac{b(c+dx)}{d(a+bx)}\right)^n} \text{Hypergeometric2F1}\left[-n, -n, 1-n, -\frac{bc-ad}{d(a+bx)}\right]$$

Program code:

```
Int[(c+d.*x_)^n_/x_,x_Symbol] :=
  -(c+d*x)^(n+1)/(c*(n+1))*Hypergeometric2F1[1,n+1,n+2,1+d*x/c] /;
FreeQ[{c,d,n},x] && Not[IntegerQ[n]]
```

```
Int[(c.+d.*x_)^n_/(a.+b.*x_),x_Symbol] :=
  -(c+d*x)^(n+1)/((n+1)*(b*c-a*d))*Hypergeometric2F1[1,n+1,n+2,TogetherSimplify[b*(c+d*x)/(b*c-a*d)]] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[n]]
```

5. $\int (a+bx)^m (c+dx)^n dx$ when $bc - ad \neq 0 \wedge m < -1$

1: $\int (a+bx)^m (c+dx)^n dx$ when $bc - ad \neq 0 \wedge m < -1 \wedge n > 0$

Reference: G&R 2.110.3 or 2.110.4 with $k = 1$

Derivation: Linear recurrence 1

Note: If $n \in \mathbb{Z}$ and $m \notin \mathbb{Z}$, there is no need to drive m toward 0 along with n .

Rule 1.1.1.2.5.1: If $bc - ad \neq 0 \wedge m < -1 \wedge n > 0$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1} (c+dx)^n}{b(m+1)} - \frac{dn}{b(m+1)} \int (a+bx)^{m+1} (c+dx)^{n-1} dx$$

Program code:

```
Int[(a_+b_*x_)^m_*(c_+d_*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+1)) -
  d*n/(b*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] && GtQ[n,0] && Not[IntegerQ[n]] && Not[IntegerQ[m]] &&
Not[ILeQ[m+n+2,0] && (FractionQ[m] || GeQ[2*n+m+1,0])] && IntLinearQ[a,b,c,d,m,n,x]
```

$$2: \int (a+bx)^m (c+dx)^n dx \text{ when } bc - ad \neq 0 \wedge m < -1$$

Reference: G&R 2.155, CRC 59a

Reference: G&R 2.110.2 or 2.110.6 with $k = 1$

Derivation: Linear recurrence 3

Derivation: Integration by parts

$$\text{Basis: } (a+bx)^m (c+dx)^n = (c+dx)^{m+n+2} \frac{(a+bx)^m}{(c+dx)^{m+2}}$$

Rule 1.1.1.2.5.2: If $bc - ad \neq 0 \wedge m < -1$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1} (c+dx)^{n+1}}{(bc-ad)(m+1)} - \frac{d(m+n+2)}{(bc-ad)(m+1)} \int (a+bx)^{m+1} (c+dx)^n dx$$

Program code:

```
Int[(a_+b_*x_)^m_*(c_+d_*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/((b*c-a*d)*(m+1)) -
  d*(m+n+2)/((b*c-a*d)*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] &&
Not[LtQ[n,-1] && (EqQ[a,0] || NeQ[c,0] && LtQ[m-n,0] && IntegerQ[n])] && IntLinearQ[a,b,c,d,m,n,x]
```

6: $\int (a+bx)^m (c+dx)^n dx$ when $bc - ad \neq 0 \wedge n > 0 \wedge m+n+1 \neq 0$

Reference: G&R 2.151, CRC 59b

Reference: G&R 2.110.1 or 2.110.5 with $k = 1$

Derivation: Linear recurrence 2

Derivation: Inverted integration by parts

Rule 1.1.1.2.6: If $bc - ad \neq 0 \wedge n > 0 \wedge m+n+1 \neq 0$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1} (c+dx)^n}{b(m+n+1)} + \frac{n(bc-ad)}{b(m+n+1)} \int (a+bx)^m (c+dx)^{n-1} dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) +
  n*(b*c-a*d)/(b*(m+n+1))*Int[(a+b*x)^m*(c+d*x)^(n-1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && GtQ[n,0] && NeQ[m+n+1,0] &&
Not[IGtQ[m,0] && (Not[IntegerQ[n]] || GtQ[m,0] && LtQ[m-n,0])] &&
Not[ILtQ[m+n+2,0]] && IntLinearQ[a,b,c,d,m,n,x]
```

$$7. \int (a+bx)^m (c+dx)^n dx \text{ when } bc - ad \neq 0 \wedge -1 < m < 0 \wedge -1 < n < 0$$

$$1. \int (a+bx)^m (c+dx)^m dx \text{ when } bc - ad \neq 0 \wedge -1 < m < 0 \wedge \text{Denominator}[m] \leq 4$$

$$1. \int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \text{ when } bc - ad \neq 0$$

$$1: \int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \text{ when } b+d = 0 \wedge a+c > 0$$

Derivation: Algebraic simplification

Basis: If $a+c > 0$, then $(a+bx)^m (c-bx)^m = ((a+bx)(c-bx))^m = (ac - b(a-c)x - b^2x^2)^m$

Rule 1.1.1.2.7.1.1.1: If $b+d = 0 \wedge a+c > 0$, then

$$\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \rightarrow \int \frac{1}{\sqrt{ac - b(a-c)x - b^2x^2}} dx$$

Program code:

```
Int[1/(Sqrt[a+_.*x_]*Sqrt[c+_.*d_.*x_]),x_Symbol] :=
  Int[1/Sqrt[a*c-b*(a-c)*x-b^2*x^2],x] /;
  FreeQ[{a,b,c,d},x] && EqQ[b+d,0] && GtQ[a+c,0]
```

$$2: \int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \text{ when } bc - ad > 0 \wedge b > 0$$

Derivation: Integration by substitution

Basis: If $b > 0$, then $\frac{1}{\sqrt{a+bx} \sqrt{c+dx}} = \frac{2}{\sqrt{b}} \text{Subst}\left[\frac{1}{\sqrt{bc-ad+dx^2}}, x, \sqrt{a+bx}\right] \partial_x \sqrt{a+bx}$

Rule 1.1.1.2.7.1.1.2: If $bc - ad > 0 \wedge b > 0$, then

$$\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \rightarrow \frac{2}{\sqrt{b}} \text{Subst} \left[\int \frac{1}{\sqrt{bc-ad+dx^2}} dx, x, \sqrt{a+bx} \right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  2/Sqrt[b]*Subst[Int[1/Sqrt[b*c-a*d+d*x^2],x],x,Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d},x] && GtQ[b*c-a*d,0] && GtQ[b,0]
```

3: $\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx$ when $bc-ad \neq 0 \wedge b-d = 0$

Derivation: Integration by substitution

Basis: If $b-d = 0$, then $\frac{1}{\sqrt{a+bx} \sqrt{c+dx}} = \frac{2}{b} \text{Subst} \left[\frac{1}{\sqrt{c-a+x^2}}, x, \sqrt{a+bx} \right] \partial_x \sqrt{a+bx}$

Rule 1.1.1.2.7.1.1.3: If $bc-ad \neq 0 \wedge b-d = 0$, then

$$\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \rightarrow \frac{2}{b} \text{Subst} \left[\int \frac{1}{\sqrt{c-a+x^2}} dx, x, \sqrt{a+bx} \right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  2/b*Subst[Int[1/Sqrt[c-a+x^2],x],x,Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && EqQ[b-d,0]
```

4: $\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx$ when $bc - ad \neq 0$

Derivation: Integration by substitution

Basis: $\frac{1}{\sqrt{a+bx} \sqrt{c+dx}} = 2 \text{Subst} \left[\frac{1}{b-dx^2}, x, \frac{\sqrt{a+bx}}{\sqrt{c+dx}} \right] \partial_x \frac{\sqrt{a+bx}}{\sqrt{c+dx}}$

Rule 1.1.1.2.7.1.1.4: If $bc - ad \neq 0$, then

$$\int \frac{1}{\sqrt{a+bx} \sqrt{c+dx}} dx \rightarrow 2 \text{Subst} \left[\int \frac{1}{b-dx^2} dx, x, \frac{\sqrt{a+bx}}{\sqrt{c+dx}} \right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  2*Subst[Int[1/(b-d*x^2),x],x,Sqrt[a+b*x]/Sqrt[c+d*x]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

$$2: \int (a+bx)^m (c+dx)^m dx \text{ when } bc - ad \neq 0 \wedge -1 < m < 0 \wedge 3 \leq \text{Denominator}[m] \leq 4$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(a+bx)^m (c+dx)^m}{((a+bx)(c+dx))^m} == 0$$

Rule 1.1.1.2.7.1.2: If $bc - ad \neq 0 \wedge -1 < m < 0 \wedge 3 \leq \text{Denominator}[m] \leq 4$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^m (c+dx)^m}{(ac + (bc+ad)x + bdx^2)^m} \int (ac + (bc+ad)x + bdx^2)^m dx$$

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^m (c+dx)^m}{((a+bx)(c+dx))^m} \int (ac + (bc+ad)x + bdx^2)^m dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  (a+b*x)^m*(c+d*x)^m/(a*c+(b*c+a*d)*x+b*d*x^2)^m*Int[(a*c+(b*c+a*d)*x+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && LeQ[3,Denominator[m],4] && AtomQ[b*c+a*d]
```

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  (a+b*x)^m*(c+d*x)^m/((a+b*x)*(c+d*x))^m*Int[(a*c+(b*c+a*d)*x+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && LeQ[3,Denominator[m],4]
```

$$2. \int (a+bx)^m (c+dx)^n dx \text{ when } bc - ad \neq 0 \wedge -1 < m < 0 \wedge m+n+1 = 0$$

$$1. \int \frac{1}{(a+bx)^{1/3} (c+dx)^{2/3}} dx \text{ when } bc - ad \neq 0$$

$$1: \int \frac{1}{(a+bx)^{1/3} (c+dx)^{2/3}} dx \text{ when } bc - ad \neq 0 \wedge \frac{d}{b} > 0$$

Rule 1.1.1.2.7.2.1.1: If $bc - ad \neq 0 \wedge \frac{d}{b} > 0$, let $q = \left(\frac{d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+bx)^{1/3} (c+dx)^{2/3}} dx \rightarrow -\frac{\sqrt{3} q}{d} \operatorname{ArcTan}\left[\frac{2q(a+bx)^{1/3}}{\sqrt{3}(c+dx)^{1/3}} + \frac{1}{\sqrt{3}}\right] - \frac{q}{2d} \operatorname{Log}[c+dx] - \frac{3q}{2d} \operatorname{Log}\left[\frac{q(a+bx)^{1/3}}{(c+dx)^{1/3}} - 1\right]$$

Program code:

```
Int[1/((a_+b_.*x_)^(1/3)*(c_+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[d/b,3]},
    -Sqrt[3]*q/d*ArcTan[2*q*(a+b*x)^(1/3)/(Sqrt[3]*(c+d*x)^(1/3))+1/Sqrt[3]] -
    q/(2*d)*Log[c+d*x] -
    3*q/(2*d)*Log[q*(a+b*x)^(1/3)/(c+d*x)^(1/3)-1] /;
    FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && PosQ[d/b]
```

2: $\int \frac{1}{(a+bx)^{1/3} (c+dx)^{2/3}} dx$ when $bc - ad \neq 0 \wedge \frac{d}{b} \neq 0$

Rule 1.1.1.2.7.2.1.2: If $bc - ad \neq 0 \wedge \frac{d}{b} \neq 0$, let $q = \left(-\frac{d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+bx)^{1/3} (c+dx)^{2/3}} dx \rightarrow \frac{\sqrt{3} q}{d} \operatorname{ArcTan}\left[\frac{1}{\sqrt{3}} - \frac{2q(a+bx)^{1/3}}{\sqrt{3}(c+dx)^{1/3}}\right] + \frac{q}{2d} \operatorname{Log}[c+dx] + \frac{3q}{2d} \operatorname{Log}\left[\frac{q(a+bx)^{1/3}}{(c+dx)^{1/3}} + 1\right]$$

Program code:

```
Int[1/((a_+b_.*x_)^(1/3)*(c_+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[-d/b,3]},
    Sqrt[3]*q/d*ArcTan[1/Sqrt[3]-2*q*(a+b*x)^(1/3)/(Sqrt[3]*(c+d*x)^(1/3))] +
    q/(2*d)*Log[c+d*x] +
    3*q/(2*d)*Log[q*(a+b*x)^(1/3)/(c+d*x)^(1/3)+1] /;
    FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && NegQ[d/b]
```

$$2: \int (a+bx)^m (c+dx)^n dx \text{ when } bc - ad \neq 0 \wedge -1 < m < 0 \wedge m+n+1 = 0$$

Derivation: Integration by substitution

Basis: If $(p \mid mp) \in \mathbb{Z} \wedge m+n+1 = 0$, then $(a+bx)^m (c+dx)^n = p \text{ Subst} \left[\frac{x^{p(m+1)-1}}{b-dx^p}, x, \frac{(a+bx)^{1/p}}{(c+dx)^{1/p}} \right] \partial_x \frac{(a+bx)^{1/p}}{(c+dx)^{1/p}}$

Rule 1.1.1.2.7.2.2: If $bc - ad \neq 0 \wedge -1 < m < 0 \wedge m+n+1 = 0$, let $p = \text{Denominator}[m]$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow p \text{ Subst} \left[\int \frac{x^{p(m+1)-1}}{b-dx^p} dx, x, \frac{(a+bx)^{1/p}}{(c+dx)^{1/p}} \right]$$

Program code:

```
Int[(a_+b_*x_)^m_*(c_+d_*x_)^n_,x_Symbol] :=
  With[{p=Denominator[m]},
    p*Subst[Int[x^(p*(m+1)-1)/(b-d*x^p),x],x,(a+b*x)^(1/p)/(c+d*x)^(1/p)] /;
    FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && EqQ[m+n+1,0]
```

$$3: \int (a+bx)^m (c+dx)^n dx \text{ when } bc - ad \neq 0 \wedge -1 < m < 0 \wedge -1 < n < 0$$

Derivation: Integration by substitution

$$\text{Basis: If } p \in \mathbb{Z}^+, \text{ then } (a+bx)^m (c+dx)^n = \frac{p}{b} \text{Subst} \left[x^{p(m+1)-1} \left(c - \frac{ad}{b} + \frac{d}{b} x^p \right)^n, x, (a+bx)^{1/p} \right] \partial_x (a+bx)^{1/p}$$

Rule 1.1.1.2.7.3: If $bc - ad \neq 0 \wedge -1 < m < 0 \wedge -1 < n < 0$, let $p = \text{Denominator}[m]$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{p}{b} \text{Subst} \left[\int x^{p(m+1)-1} \left(c - \frac{ad}{b} + \frac{d}{b} x^p \right)^n dx, x, (a+bx)^{1/p} \right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  With[{p=Denominator[m]},
    p/b*Subst[Int[x^(p*(m+1)-1)*(c-a*d/b+d*x^p/b)^n,x,(a+b*x)^(1/p)]] /;
    FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && LtQ[-1,n,0] && LeQ[Denominator[n],Denominator[m]] &&
    IntLinearQ[a,b,c,d,m,n,x]
```

$$8: \int (a+bx)^m (c+dx)^n dx \text{ when } bc - ad \neq 0 \wedge m+n+2 \in \mathbb{Z}^- \wedge m \neq -1$$

Reference: G&R 2.155, CRC 59a

Reference: G&R 2.110.2 or 2.110.6 with $k = 1$

Derivation: Linear recurrence 3

Derivation: Integration by parts

$$\text{Basis: } (a+bx)^m (c+dx)^n = (c+dx)^{m+n+2} \frac{(a+bx)^m}{(c+dx)^{m+2}}$$

Rule 1.1.1.2.8: If $bc - ad \neq 0 \wedge m+n+2 \in \mathbb{Z}^- \wedge m \neq -1$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1} (c+dx)^{n+1}}{(bc-ad)(m+1)} - \frac{d(m+n+2)}{(bc-ad)(m+1)} \int (a+bx)^{m+1} (c+dx)^n dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/((b*c-a*d)*(m+1)) -
  d*Simplify[m+n+2]/((b*c-a*d)*(m+1))*Int[(a+b*x)^Simplify[m+1]*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && ILtQ[Simplify[m+n+2],0] && NeQ[m,-1] &&
(SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]])
```

H. $\int (a+bx)^m (c+dx)^n dx$ when $bc-ad \neq 0$

1. $\int (bx)^m (c+dx)^n dx$

1: $\int (bx)^m (c+dx)^n dx$ when $m \notin \mathbb{Z} \wedge (n \in \mathbb{Z} \vee c > 0)$

Rule 1.1.1.2.H.1.1: If $m \notin \mathbb{Z} \wedge (n \in \mathbb{Z} \vee c > 0)$, then

$$\int (bx)^m (c+dx)^n dx \rightarrow \frac{c^n (bx)^{m+1}}{b(m+1)} \text{Hypergeometric2F1}\left[-n, m+1, m+2, -\frac{dx}{c}\right]$$

Program code:

```
Int[(b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  c^n*(b*x)^(m+1)/(b*(m+1))*Hypergeometric2F1[-n,m+1,m+2,-d*x/c] /;
FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[m]] && (IntegerQ[n] || GtQ[c,0] && Not[EqQ[n,-1/2]] && EqQ[c^2-d^2,0] && GtQ[-d/(b*c),0])
```

2: $\int (bx)^m (c+dx)^n dx$ when $n \notin \mathbb{Z} \wedge (m \in \mathbb{Z} \vee -\frac{d}{bc} > 0)$

Rule 1.1.1.2.H.1.2: If $n \notin \mathbb{Z} \wedge (m \in \mathbb{Z} \vee -\frac{d}{bc} > 0)$, then

$$\int (bx)^m (c+dx)^n dx \rightarrow \frac{(c+dx)^{n+1}}{d(n+1) \left(-\frac{d}{bc}\right)^m} \text{Hypergeometric2F1}\left[-m, n+1, n+2, 1+\frac{dx}{c}\right]$$

Program code:

```
Int[(b_*x_)^m_*(c_+d_*x_)^n_,x_Symbol] :=
  (c+d*x)^(n+1) / (d*(n+1) * (-d/(b*c))^m) *Hypergeometric2F1[-m,n+1,n+2,1+d*x/c] /;
  FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[n]] && (IntegerQ[m] || GtQ[-d/(b*c),0])
```

$$3. \int (bx)^m (c+dx)^n dx \text{ when } m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge c \neq 0 \wedge -\frac{d}{bc} \neq 0$$

$$1: \int (bx)^m (c+dx)^n dx \text{ when } m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge c \neq 0 \wedge -\frac{d}{bc} \neq 0 \wedge (m \in \mathbb{R} \vee n \in \mathbb{R})$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(c+dx)^n}{\left(1+\frac{dx}{c}\right)^n} = 0$$

Rule 1.1.1.2.H.1.3.1: If $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge c \neq 0 \wedge -\frac{d}{bc} \neq 0 \wedge (m \in \mathbb{R} \vee n \in \mathbb{R})$, then

$$\int (bx)^m (c+dx)^n dx \rightarrow \frac{c^{\text{IntPart}[n]} (c+dx)^{\text{FracPart}[n]}}{\left(1+\frac{dx}{c}\right)^{\text{FracPart}[n]}} \int (bx)^m \left(1+\frac{dx}{c}\right)^n dx$$

Program code:

```
Int[(b_*x_)^m_*(c_+d_*x_)^n_,x_Symbol] :=
  c^IntPart[n] * (c+d*x)^FracPart[n] / (1+d*x/c)^FracPart[n] * Int[(b*x)^m * (1+d*x/c)^n, x] /;
  FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[GtQ[c,0]] && Not[GtQ[-d/(b*c),0]] &&
  (RationalQ[m] && Not[EqQ[n,-1/2]] && EqQ[c^2-d^2,0]) || Not[RationalQ[n]]
```

$$2: \int (bx)^m (c+dx)^n dx \text{ when } m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge c \neq 0 \wedge -\frac{d}{bc} \neq 0 \wedge \neg (m \in \mathbb{R} \vee n \in \mathbb{R})$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(bx)^m}{\left(-\frac{dx}{c}\right)^m} = 0$$

Rule 1.1.1.2.H.1.3.2: If $m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge c \neq 0 \wedge -\frac{d}{bc} \neq 0 \wedge \neg (m \in \mathbb{R} \vee n \in \mathbb{R})$, then

$$\int (bx)^m (c+dx)^n dx \rightarrow \frac{\left(-\frac{bc}{d}\right)^{\text{IntPart}[m]} (bx)^{\text{FracPart}[m]}}{\left(-\frac{dx}{c}\right)^{\text{FracPart}[m]}} \int \left(-\frac{dx}{c}\right)^m (c+dx)^n dx$$

Program code:

```
Int[(b_*x_)^m_*(c_+d_*x_)^n_,x_Symbol] :=
  (-b*c/d)^IntPart[m]*(b*x)^FracPart[m]/(-d*x/c)^FracPart[m]*Int[(-d*x/c)^m*(c+d*x)^n,x] /;
  FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[GtQ[c,0]] && Not[GtQ[-d/(b*c),0]]
```

$$2. \int (a+bx)^m (c+dx)^n dx \text{ when } bc-ad \neq 0 \wedge m \notin \mathbb{Z}$$

$$1: \int (a+bx)^m (c+dx)^n dx \text{ when } bc-ad \neq 0 \wedge m \notin \mathbb{Z} \wedge \left(n \in \mathbb{Z} \vee \frac{b}{bc-ad} > 0 \right)$$

Rule 1.1.1.2.H.2.2.1: If $bc-ad \neq 0 \wedge m \notin \mathbb{Z} \wedge \left(n \in \mathbb{Z} \vee \frac{b}{bc-ad} > 0 \right)$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(a+bx)^{m+1}}{b(m+1) \left(\frac{b}{bc-ad} \right)^n} \text{Hypergeometric2F1} \left[-n, m+1, m+2, -\frac{d(a+bx)}{bc-ad} \right]$$

Program code:

```
Int[(a+_b_.*x_)^m_*(c+_d_.*x_)^n_,x_Symbol] :=
  (b*c-a*d)^n*(a+b*x)^(m+1)/(b^(n+1)*(m+1))*Hypergeometric2F1[-n,m+1,m+2,-d*(a+b*x)/(b*c-a*d)] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[m]] && IntegerQ[n]
```

```
Int[(a+_b_.*x_)^m_*(c+_d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)/(b*(m+1)*(b/(b*c-a*d))^n)*Hypergeometric2F1[-n,m+1,m+2,-d*(a+b*x)/(b*c-a*d)] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[b/(b*c-a*d),0] &&
  (RationalQ[m] || Not[RationalQ[n] && GtQ[-d/(b*c-a*d),0]])
```

$$2: \int (a+bx)^m (c+dx)^n dx \text{ when } bc-ad \neq 0 \wedge m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge \frac{b}{bc-ad} \neq 0$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{(c+dx)^n}{\left(\frac{bc}{bc-ad} + \frac{bdx}{bc-ad}\right)^n} = 0$$

Rule 1.1.1.2.H.2.2.2: If $bc-ad \neq 0 \wedge m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge \frac{b}{bc-ad} \neq 0$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{(c+dx)^{\text{FracPart}[n]}}{\left(\frac{b}{bc-ad}\right)^{\text{IntPart}[n]} \left(\frac{b(c+dx)}{bc-ad}\right)^{\text{FracPart}[n]}} \int (a+bx)^m \left(\frac{bc}{bc-ad} + \frac{bdx}{bc-ad}\right)^n dx$$

Program code:

```
Int[(a+b*x)^m*(c+d*x)^n,x_Symbol] :=
(c+d*x)^FracPart[n]/((b/(b*c-a*d))^IntPart[n]*(b*(c+d*x)/(b*c-a*d))^FracPart[n])*
Int[(a+b*x)^m*Simp[b*c/(b*c-a*d)+b*d*x/(b*c-a*d),x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && (RationalQ[m] || Not[SimplerQ[n+1,m+1]])
```


S: $\int (a+bx)^m (c+dx)^n dx$ when $u = e+fx$

Derivation: Integration by substitution

Rule 1.1.1.2.S: If $u = e+fx$, then

$$\int (a+bx)^m (c+dx)^n dx \rightarrow \frac{1}{f} \text{Subst} \left[\int (a+bx)^m (c+dx)^n dx, x, u \right]$$

Program code:

```
Int[(a_.+b_.*u_)^m_.*(c_.+d_.*u_)^n_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x)^m*(c+d*x)^n,x],x,u] /;
FreeQ[{a,b,c,d,m,n},x] && LinearQ[u,x] && NeQ[Coefficient[u,x,0],0]
```

```
(* IntLinearQ[a,b,c,d,m,n,x] returns True iff (a+b*x)^m*(c+d*x)^n is integrable wrt x in terms of non-hypergeometric functions. *)
IntLinearQ[a_,b_,c_,d_,m_,n_,x_] :=
  IGtQ[m,0] || IGtQ[n,0] || IntegersQ[3*m,3*n] || IntegersQ[4*m,4*n] || IntegersQ[2*m,6*n] || IntegersQ[6*m,2*n] || ILtQ[m+n,-1] || IntegerQ[m+n]
```